Demo alkalmazás – Osztályok, Interfészek

A „vechicle” az osztály és kivétel osztály nevében csúnya elgépelés található, helyesen vehicle. Ezt a jegyzetben egyelőre nem javítottam! Az osztályok nevét leszámítva ezt a jegyzetben már javítottam.

Interface „Startable”

INTERFACE zmk_if_startable
  PUBLIC .
    DATA state TYPE string READ-ONLY.
    METHODS start.
    METHODS stop.
ENDINTERFACE.
Generált kivétel osztály
CLASS zcx_vechicle DEFINITION
  PUBLIC
  INHERITING FROM cx_static_check
  FINAL
  CREATE PUBLIC .

  PUBLIC SECTION.
    INTERFACES if_t100_message .
    INTERFACES if_t100_dyn_msg .

    METHODS constructor
      IMPORTING
        !textid   LIKE if_t100_message=>t100key OPTIONAL
        !previous LIKE previous OPTIONAL .
  PROTECTED SECTION.
  PRIVATE SECTION.
ENDCLASS.

CLASS zcx_vechicle IMPLEMENTATION.
  METHOD constructor ##ADT_SUPPRESS_GENERATION.
    super->constructor(
      previous = previous
    ).
    CLEAR me->textid.
    IF textid IS INITIAL.
      if_t100_message~t100key = if_t100_message=>default_textid.
    ELSE.
      if_t100_message~t100key = textid.
    ENDIF.
  ENDMETHOD.
ENDCLASS.

Pedestrian osztály
CLASS zmk_cl_pedestrian DEFINITION
  PUBLIC
  FINAL
  CREATE PUBLIC .

  PUBLIC SECTION.
    INTERFACES zmk_if_startable.
  PROTECTED SECTION.
  PRIVATE SECTION.
ENDCLASS.

CLASS zmk_cl_pedestrian IMPLEMENTATION.
  METHOD zmk_if_startable~start.
    zmk_if_startable~state = 'PEDESTRIAN MOVING'.
  ENDMETHOD.
  METHOD zmk_if_startable~stop.
    zmk_if_startable~state = 'PEDESTRIAN STAYS'.
  ENDMETHOD.
ENDCLASS.

Vehicle osztály
CLASS zmk_cl_vechicle DEFINITION
  PUBLIC
  CREATE PUBLIC .

  PUBLIC SECTION.
    INTERFACES zmk_if_startable.

    " Static attribute used to count how many times
    " any constructor of the class of its subclasses has been
    " called
    CLASS-DATA number_of_vechicle TYPE i READ-ONLY.

    " Public constructor
    METHODS constructor
      IMPORTING
                iv_colour    TYPE string
                iv_max_speed TYPE i
      RAISING   zcx_vechicle.

    " Setter/Getter for current speed
    METHODS set_current_speed
      IMPORTING
                iv_current_speed TYPE i
      RAISING   zcx_vechicle.

    METHODS get_current_speed
      RETURNING VALUE(rv_current_speed) TYPE i.

    " Setter/Getter for colour
    " For demonstration purpose getter uses export parameter
    METHODS set_colour
      IMPORTING
        iv_colour TYPE string.

    METHODS get_colour
      EXPORTING
        ev_colour TYPE string.

    " Getter for max speed
    METHODS get_max_speed
      RETURNING VALUE(rv_max_speed) TYPE i.

  PROTECTED SECTION.
  PRIVATE SECTION.
    " Example private, member attributes
    DATA: max_speed     TYPE i,
          current_speed TYPE i,
          colour        TYPE string.
ENDCLASS.

CLASS zmk_cl_vechicle IMPLEMENTATION.
  METHOD constructor.
    " Get over colour from input
    colour = iv_colour.

    " Check max speed settings
    IF iv_max_speed GE 0.
      max_speed = iv_max_speed.
    ELSE.
      RAISE EXCEPTION TYPE zcx_vechicle.
    ENDIF.
    number_of_vechicle += 1.

    " Set default status as "Stopped"
    zmk_if_startable~state = 'Stopped'.
  ENDMETHOD.

  METHOD set_colour.
    colour = iv_colour.
  ENDMETHOD.

  METHOD get_colour.
    ev_colour = colour.
  ENDMETHOD.

  METHOD set_current_speed.
    IF iv_current_speed GT max_speed
     OR iv_current_speed LT 0.
      RAISE EXCEPTION TYPE zcx_vechicle.
    ENDIF.

    current_speed = iv_current_speed.
  ENDMETHOD.

  METHOD  get_current_speed.
    rv_current_speed = current_speed.
  ENDMETHOD.

  METHOD get_max_speed.
    rv_max_speed = max_speed.
  ENDMETHOD.

  METHOD zmk_if_startable~start.
    zmk_if_startable~state = 'Started'.
  ENDMETHOD.

  METHOD zmk_if_startable~stop.
    zmk_if_startable~state = 'Stopped'.
  ENDMETHOD.
ENDCLASS.

Car osztály
CLASS zmk_cl_car DEFINITION
  PUBLIC
  FINAL
  CREATE PUBLIC
  INHERITING FROM zmk_cl_vechicle.

  PUBLIC SECTION.
  PROTECTED SECTION.
  PRIVATE SECTION.
ENDCLASS.

CLASS zmk_cl_car IMPLEMENTATION.
ENDCLASS.

Truck osztály

CLASS zmk_cl_truck DEFINITION
  PUBLIC
  FINAL
  CREATE PUBLIC
  INHERITING FROM zmk_cl_vechicle.

  PUBLIC SECTION.
    METHODS constructor
      IMPORTING
        iv_max_speed      TYPE i
        iv_max_cargo_size TYPE i OPTIONAL
    RAISING zcx_vechicle.

    METHODS set_max_cargo_size
      IMPORTING
        iv_max_cargo_size TYPE i.

    METHODS get_max_cargo_size
      RETURNING VALUE(rv_max_cargo_size) TYPE i.

  PRIVATE SECTION.
    DATA max_cargo_size TYPE i.
ENDCLASS.

CLASS zmk_cl_truck IMPLEMENTATION.
  METHOD constructor.
    " By design all truck is Grey
    super->constructor(
        iv_colour    = 'Grey'
        iv_max_speed = iv_max_speed ).

    set_max_cargo_size( iv_max_cargo_size = iv_max_cargo_size ).

  ENDMETHOD.

  METHOD set_max_cargo_size.
    max_cargo_size = iv_max_cargo_size.
  ENDMETHOD.

  METHOD get_max_cargo_size.
    rv_max_cargo_size = max_cargo_size.
  ENDMETHOD.
ENDCLASS.




Egyszerű, különböző használati eseteket bemutató alkalmazás

CLASS zmk_cl_vechicle_tester_1 DEFINITION
  PUBLIC
  FINAL
  CREATE PUBLIC .

  PUBLIC SECTION.

    INTERFACES if_oo_adt_classrun .
  PROTECTED SECTION.
  PRIVATE SECTION.
ENDCLASS.

CLASS zmk_cl_vechicle_tester_1 IMPLEMENTATION.
  METHOD if_oo_adt_classrun~main.

    DATA lo_vehicle   TYPE REF TO zmk_cl_vechicle.
    DATA lo_truck     TYPE REF TO zmk_cl_truck.
    DATA lv_colour    TYPE string.              " Used for importing parameter

    DATA lt_vehicle   TYPE TABLE OF REF TO zmk_cl_vechicle.
    DATA lo_startable TYPE REF TO zmk_if_startable.

    "   ---------- First Vehicle ----
    TRY.
        " First we try to create a Red vehicle
        " Creation is based on a known, exact class
        CREATE OBJECT lo_vehicle
          EXPORTING
            iv_colour    = 'Red'
            iv_max_speed = 100.

        " We address the known interface to "start" it
        lo_vehicle->zmk_if_startable~start( ).

        " To check getter we call method (old fashioned way)
        CALL METHOD lo_vehicle->get_colour
          IMPORTING
            ev_colour = lv_colour.
        out->write( lv_colour ).

        " To check how many car instance has been created, we call the static method
        " (new fashion)
        out->write( zmk_cl_vechicle=>number_of_vechicle ).

        " To keep the references, we add it to an internal table
        " where references to vehicles are stored
        APPEND lo_vehicle TO lt_vehicle.

      CATCH zcx_vechicle.
        out->write( 'Cannot create this vechicle' ).
    ENDTRY.

    "   ---------- Negative test ----
    TRY.
        " Try to create a Red vehicle with negative speed
        CREATE OBJECT lo_vehicle
          EXPORTING
            iv_colour    = 'Red'
            iv_max_speed = -10.

      CATCH zcx_vechicle.
        out->write( 'Cannot create this vechicle' ).
    ENDTRY.

    " ---------- Second Vehicle ----
    TRY.
        CREATE OBJECT lo_vehicle
          EXPORTING
            iv_colour    = 'Blue'
            iv_max_speed = 50.

        " Take care that the lo_vehicle reference variable is now
        " pointing to the newly created instance
        " If the lt_vehicle table would not have a reference to the
        " previously created instance, the instance would "lost" and the
        " garbage collector would free up the memory later

        out->write( zmk_cl_vechicle=>number_of_vechicle ).

        " To keep the references, we add it to an internal table
        " where references to vehicles are stored
        APPEND lo_vehicle TO lt_vehicle.

      CATCH zcx_vechicle.
        out->write( 'Cannot create this vechicle' ).
    ENDTRY.

    " ---------- First truck ----
    TRY.
        CREATE OBJECT lo_truck
          EXPORTING
            iv_max_speed      = 80
            iv_max_cargo_size = 100.

        " Call method (new fasion)
        lo_truck->get_colour(
          IMPORTING
            ev_colour = lv_colour ).

        out->write( lv_colour ).

        out->write( zmk_cl_vechicle=>number_of_vechicle ).

        APPEND lo_truck TO lt_vehicle.

      CATCH zcx_vechicle.
        out->write( 'Cannot create this vechicle' ).
    ENDTRY.

    "-----------------------------------------------------------
    " Process internal table pointing to references in a loop
    "-----------------------------------------------------------
    out->write( 'Mass processing of reference table' ).

    LOOP AT lt_vehicle INTO lo_vehicle.
      lo_vehicle->get_colour(
        IMPORTING
          ev_colour = lv_colour ).
      out->write( lv_colour ).

      " Example of down-casting with trial
      " Not all vehicle is a truck, so we check if cast is successful
      " or not
      TRY.
          lo_truck ?= lo_vehicle.       " Test of down-cast

          out->write( lo_truck->get_max_cargo_size( ) ).
        CATCH cx_sy_move_cast_error.
          " Do nothing, this is not a truck
      ENDTRY.

      " Accessing vehicle as "Startable" (interface)
      lo_startable = lo_vehicle.

      " Writ the state of the vehicle as accessing the public, read-only member attribute
      out->write( lo_startable->state ).

    ENDLOOP.

  ENDMETHOD.
ENDCLASS.




Egyszerű, down-cast alkalmazást bemutató konzol alkalmazás
CLASS zmk_cl_startable_tester_1 DEFINITION
  PUBLIC
  FINAL
  CREATE PUBLIC .

  PUBLIC SECTION.

    INTERFACES if_oo_adt_classrun .
  PROTECTED SECTION.
  PRIVATE SECTION.
ENDCLASS.

CLASS zmk_cl_startable_tester_1 IMPLEMENTATION.
  METHOD if_oo_adt_classrun~main.
    DATA: lt_startable TYPE TABLE OF REF TO zmk_if_startable,
          lo_startable TYPE REF TO zmk_if_startable.

    " Create reference variable referencing to an interface
    " In this case during instantiation the implementating class shall be named
    CREATE OBJECT lo_startable TYPE zmk_cl_pedestrian.
    APPEND lo_startable TO lt_startable.

    " Recreate the reference variable with a different implementation
    CREATE OBJECT lo_startable TYPE zmk_cl_vechicle
      EXPORTING
        iv_colour     = 'Red'
        iv_max_speed = 100.
    APPEND lo_startable TO lt_startable.

    " Process the internal table pointing to the reference table
    " Take care, that from this logic point of view it does not matter
    " how the exact implementation of the signature looks like
    LOOP AT lt_startable INTO lo_startable.
      lo_startable->start( ).
      out->write( lo_startable->state ).
    ENDLOOP.

  ENDMETHOD.
ENDCLASS.

